AD-A091 992 ARMY COMBINED ARMS COMBAT DEVELOPMENTS ACTIVITY FORT--ETC F/G 15/7

AN HP-67 COMBAT MODEL.(U)

JUL 77 R E DAVIS

UNCLASSIFIED CACDA-TR-8-77 NL

END

DATE FILMED

1-81

DTIC

LEVEL II

Technical Paper TP 8-77
Jul[redacted]77

CACDA-TR-8-77

Directorate of Combat Operations Analysis
US Army Combined Arms Combat Developments Activity
Fort Leavenworth, Kansas 66027

AN HP-67 COMBAT MODEL.

by

COL Reed E. Davis, Jr
Director
Combat Operations Analysis

ABSTRACT

Using a common data base and a few inputs during execution, this HP-67 Combat Model determines firers, targets, engagements, results, data base updates, and specified output for each time step of a tank-antitank battle. This base version represents only tank and antitank missiles on each side. However, the power of the HP-67 will accommodate expansions; e.g., artillery, mines, cannon-launched guided projectiles (CLGP), attack helicopters, and air defenses.

Model mathematics are presented, and the model macroflow and overlay structure are described. An operators guide is included along with program listings for the 7 submodels and 20 subroutines.

ACKNOWLEDGEMENTS

The advice and assistance provided by Mr Richard C. Rinkel and Mrs Elizabeth Etheridge in the preparation of this paper were welcome and helpful. I am grateful for their experienced counsel and willing help. Thanks also to Ms Mary Slead for her excellent typing support during the development of the paper. Responsibility for this early attempt to represent small unit armored combat on an HP-67, of course, rests with me.

TABLE OF CONTENTS

Technical Paper TP 8-77

TABLE OF CONTENTS (concluded)

LIST OF FIGURES

CHAPTER 1
INTRODUCTION

1-1. PURPOSE. This paper describes an HP-67 Combat Model that simulates battalion level tank/antitank combat. The model includes representation of terrain/line of sight, target detection and acquisition, depth of opposing force formations, fire allocation, weapon engagement rates, suppression of antitank guided missile (ATGM) systems, kill computations, and the commitment of reserve forces by either side. In the base version, only tank and antitank weapons are simulated; however, the power of the HP-67 will accommodate expansions; e.g., the play of artillery, cannon-launched guided projectiles (CLGP), mines, attack helicopters, and air defense.

1-2. ORGANIZATION. Following this introductory chapter, the second chapter presents the mathematics of the model in the following sequence.

- Terrain/line of sight
- Probability of target detection/acquisition
- Targets acquired
- Engagement rates
- Suppression
- Number of firers
- Single shot hit probabilities
- Targets hit
- Allocation of hits
- Kill calculations
- Combat status output
- Reserve commitment

Chapter 3 of the paper illustrates the flow of model computations and explains the overlay, submodel, and subroutine structure of the model. Chapter 4 is an operators guide, and chapter 5 contains the program listings for the model.

CHAPTER 2
MODEL MATHEMATICS

2-1. GENERAL. The model is time- and range-increment sequenced; all the processes outlined in chapter 1 are executed each step with the exception of data update and reserve commitment, which are at the user's option. The following paragraphs describe the mathematics of the processes.

2-2. TERRAIN/LINE OF SIGHT. Probability of line of sight (PLOS) between firer and target is:

$$P(LOS) = (a + 1.68\ r/\bar{r})\ e^{-1.68\ r/\bar{r}} \tag{1}$$

where:

r = battle range

$\bar{r}$ = a terrain parameter describing generally the effects of terrain on LOS. (When $r = \bar{r}$, the P(LOS) = 0.5 in uncluttered terrain.)

a = a terrain parameter describing the amount of close-in clutter. (In fully uncluttered terrain, $a = 1$; in the vicinity of Fulda, $a \simeq 0.57$., This statistic measures only the probability of LOS as a function of range between two random locations in a given terrain. Thus, the effects of LOS duration must be taken into account in determining operational engagement rates.)

Equation 1 is based on a DIVWAG formulation (reference 1) as modified by the author on the basis of TETAM Phase I Europe results (reference 2).

2-3. PROBABILITY OF TARGET DETECTION/ACQUISITION. Both random and cued detections are represented by a single equation. Differences in these two detection phenomena are represented in the allocation of fire calculations (see paragraph 2-10). The time to detect is assumed to be distributed exponentially. Thus, probability of detection is represented as:

$$p_i = 1 - e^{-\lambda_i \Delta t} \tag{2}$$

where:

λ_i = current detection rate for the i^{th} weapon system. (λ_i can be changed for the i^{th} weapon type prior to any time step by calling submodel update (overlay 4). Thus, the effects of range, smoke, fog, etc. can be represented.)

Δt = the time step

2-4. TARGETS ACQUIRED. The number of targets acquired (T_a) by a weapon system is simply:

$$T_a = PLOS \cdot p_i \cdot N(t) \qquad (3)$$

where PLOS and p_i are as defined above and N(t) is the number of targets alive on the opposing side (B(t) for Blue, R(t) for Red).

2-5. ENGAGEMENT RATES.

a. ATGM Engagement Rates. ATGM operational engagement rates are a function of acquisition rates, missile reload times, times of flight, terrain abort rates, and suppression effects. Acquisition rates are represented explicitly in the model, the input maximum engagement rates account for terrain aborts, and the explicit representation of suppression is as discussed in paragraph 2-6 below. Thus, the missile reload and flight times remain and can be represented as a linear function of range. Thus:

$$E_i(r) = a_i - b_i\, r \qquad (4)$$

where r is the range and a_i and b_i are positive fitting parameters for the i^{th} ATGM.

b. Tank Engagement Rates. As the battle closes tanks do not enter the battle instantaneously. The defender's tanks are deployed in some depth, and the attacker's tanks are part of a selected attack formation. Thus, the tanks of each adversary enter the battle, as it closes, over a period of time that is largely a function of deployed depth. The average engagement rate for each attacker tank is based on the assumption that his tanks arrive within effective range exponentially as a function of formation depth. Thus:

$$0 \qquad X_i > X_0 \qquad (5)$$

$$P' = e^{\frac{-\ln(p+1)(X_i - X_0)}{d}} - 1 \qquad X_0 - d \le X_i \le X_0 \qquad (6)$$

$$p \qquad X_i < X_0 - d \qquad (7)$$

The average engagement rate for each defender tank is based on the assumption that his tanks are deployed uniformly in depth. Thus:

$$0 \qquad X_i > X_0 \tag{8}$$

$$p' = \frac{p}{d}(X_0 - X_i) \qquad X_0 - d \leq X_i \leq X_0 \tag{9}$$

$$p \qquad X_i < X_0 - d \tag{10}$$

where:

p = individual tank operational engagement rate

d = formation depth

X_0 = effective opening range

X_i = battle range

2-6. SUPPRESSION. Previous attempts to represent suppressive effects have met with limited success due to a lack of understanding the phenomena. Operative HP-67 coding of the Litton Suppression Model (reference 3) is provided as an option for representing the conditional suppression of ATGMs by direct tank fire. The percent suppression is:

$$S = \frac{e^{\beta}}{e^{\beta} + 1} \tag{11}$$

where:

$$\beta = 10e^{-.04\,(1 - f)^2/\sigma f} - 5 \tag{12}$$

where:

f = fractional ATGM losses the last Δt

σ = a suppression parameter.

Therefore:

$$N' = N\ (1 - S) \tag{13}$$

where:

N = number of potential firers

S = percent suppression from equation 11

N' = number of effective firers.

2-7. NUMBER OF FIRERS. The number of firers is at any given time step (t) of the model simply:

$$\text{Blue ATGMs} = \alpha_1 B(t) \tag{14}$$

$$\text{Red ATGMs} = \alpha_2 R(t) \tag{15}$$

$$\text{Blue Tanks} = (1 - \alpha_1)\ B(t) \tag{16}$$

$$\text{Red Tanks} = (1 - \alpha_2)\ R(t) \tag{17}$$

where:

B(t) = total Blue survivors

R(t) = total Red survivors

α_i = current ratio of ATGMs to total systems.

If the suppression option is exercised the firer numbers from equations 15 and 17 are modified by (1 - S) as in equation 13.

2-8. SINGLE SHOT HIT PROBABILITIES (SSHP).

a. Tank SSHP Calculations. In this version of the model, tank SSHPs are a quadratic function of range. Good fits are obtained if the tank open fire rule is that the first round hit probability ≥ 0.3; otherwise, tank SSHPs are a cubic function of range. Thus:

$$SSHP_i\ (r) = 1 + a_i r + b_i r^2 \tag{18}$$

or:

$$SSHP_i\ (r) = 1 + a_i r + b_i r^2 + C_i r^3 \qquad (19)$$

where r is the range and a_i, b_i, and c_i are fitting parameters, usually but not always negative in sign, for the i^{th} tank.

b. AT Missile SSHP Calculation. Current AT missile system hit probabilities behave logarithmically as a function of range. Thus, the SSHP for any system is:

$$SSHP_i\ (r) = a_i + b_i \ln r \qquad (20)$$

where r is the range and a_i and b_i are positive fitting parameters for the i^{th} ATGM.

2-9. TARGETS HIT. Numbers of targets hit by each weapon in each time step is calculated as in the Jiffy Game (reference 4):

$$H = T_a\ (1 - (1 - p/T_a)^{nr\Delta t}) \qquad (21)$$

where:

H = number of hits, computed sequentially for each Red and Blue weapon, against targets acquired and engaged during a single time step.

T_a = targets acquired and engaged; calculated from targets available and the probability of acquisition based on terrain, range, and weapon system detection capabilities.

p = SSHP for each class of firers as a function of range against a target of average aspect angle and dimensions and specified cover. SSHPs are calculated for each time step during execution.

n = number of firers, calculated for each weapon type and each time step during execution.

r = engagement rate for type firer, calculated for each weapon type and each time step during execution.

Δt = 1 minute in base version.

2-10. ALLOCATION OF HITS BETWEEN TANK AND ATGM TARGETS. Two principal factors that influence the allocation of fires are the firer's preference for targets by type and the instantaneous rate of cued detections for particular type targets. Initially it will be assumed that the firer will satisfy his preferences within the constraint imposed by the number of cued detections. This constraint is used along with user preferences to determine the parameters for an allocation function (see figure 2-1).



Figure 2-1. Allocation (y) of fire to hard tank targets

The equation is of the form:

$$y = a_1x^3 + a_2x^2 + a_3x + a_4 \quad (22)$$

or:

$$a_1x^3 + a_2x^2 + a_3x = y - 1 \quad (23)$$

From input data three simultaneous equations are written:

$$a_1x_0^3 + 2\,a_2x_0^2 + 8\,a_3x_0 = 8\,(f_1 - 1) \quad (24)$$

$$a_1x_0^3 + a_2x_0^2 + a_3x_0 = -\,\alpha \quad (25)$$

$$a_1x_r^{\ 3} + a_2x_r^{\ 2} + a_3x_r = f_2 - 1 \qquad (26)$$

where:

x_0 = opposing tank open fire range

x_r = initial battle range

α = ratio of opposing soft targets to total targets

f_2, f_1 = user preferences within established constraints on cued versus random detections and must be determined from available data on random versus cued target detections.

These three simultaneous equations are solved for a_1, a_2, and a_3 and the resulting equations (22) coded into the model.

2-11. KILL CALCULATIONS. Kills are computed by target type by entering a probability of kill given a hit ($P_{K/H}$) when the program halts after calculating hits by target type and multiplying $P_{K/H}$ times number of hits.

2-12. COMBAT STATUS OUTPUT. Status data are output at each time step of the program using the following formulae:

$$\text{Percent Blue survivors} = B(t)/B_0 = 1 - \frac{\Sigma\, b(t)}{B_0} \qquad (27)$$

$$\text{Percent Red survivors} = R(t)/R_0 = 1 - \frac{\Sigma\, r(t)}{R_0} \qquad (28)$$

$$\text{Survivor force ratio difference (SFRD)} = B(t)/B_0 - R(t)/R_0 \qquad (29)$$

$$\text{Loss exchange ratio (LER)} = \frac{\Sigma r(t)}{\Sigma b(t)} \qquad (30)$$

$$\text{Fractional exchange ratio (FER)} = \frac{LER}{R_0/B_0} \qquad (31)$$

where:

$B(t)$ = current Blue survivors

$R(t)$ = current Red survivors

B_0 = initial Blue strength

R_0 = initial Red strength

$\Sigma b(t)$ = cumulative Blue losses

$\Sigma r(t)$ = cumulative Red losses

2-13. RESERVE COMMITMENT. Commitment of reserves is handled by:

$$\alpha'_2 = \frac{\alpha_2 R(t) + \alpha_{\Delta 2} \Delta R(t)}{R(t) + \Delta R(t)} \tag{32}$$

where:

α'_2 = updated ratio of Red ATGMs to total Red

α_2 = ratio of Red ATGMs to total Red prior to commitment

$\alpha_{\Delta 2}$ = ratio of Red ATGMs to total Red in reserve

$R(t)$ = total Red prior to commitment

$\Delta R(t)$ = total Red in reserve

The same equation is used to compute α'_1 for Blue reserve commitment.

2-14. REFERENCES.

1. TR 8-76, Division War Game (DIVWAG) Model Documentation, Volume II, Programmer/Analyst Manual, Part 3. Computer Sciences Corporation, as updated by Combat Operations Analysis Directorate, Fort Leavenworth, Kansas 66027, July 1976.

2. TM 2-73, TETAM Effectiveness Evaluation, Phase I. Combat Operations Analysis Directorate, Fort Leavenworth, Kansas 66027, November 1973.

3. TR 73/002, Relationship of Supporting Weapon Systems Performance Characteristics to Suppression of Individuals and Small Units. Litton Systems, Inc. Ralph P. Winter and E. Robert Clovis, January 1973.

4. TR 2-77, CACDA Jiffy War Game Technical Manual, Combat Operations Analysis Directorate, Fort Leavenworth, Kansas 66027, June 1977.

CHAPTER 3
MODEL FLOW AND OVERLAY STRUCTURE

3-1. MODEL FLOW. The macroflow of the HP-67 Combat Model is shown in figure 3-1.

3-2. OVERLAY STRUCTURE. This HP-67 Combat Model consists of four overlays. Using a common data base and a few inputs during execution, seven submodels and twenty subroutines determine firers, targets, engagements, results, data base updates, and specified output for each time step of a tank-antitank battle. Figures 3-2 through 3-5 show the structure of the four overlays.



Figure 3-1. HP-67 Combat Model macroflow chart

Overlay 1

Submodel A - Calls subroutines 0 and 1.x and calculates targets hit by Blue or Red ATGMs on return.

Submodel B - Allocates fires and calculates attrition. Calls subroutines 2.y, 5.x, and 6.x for execution control.

Subroutine 0 - Calls subroutine 7.x to calculate the number of firers and targets acquired. Calls subroutine 8.y to calculate the SSHP and engagement rate for Blue AT missiles.

Subroutine 1.x - Calls subroutine 7.x to calculate the number of firers and targets acquired. Calls subroutine 8.y to calculate the SSHP and engagement rate for Red AT missiles.

Subroutine 2.y - Determines next weapon code and next overlay when called.

Subroutine 3.y - Determines SSHP and engagement rate calculation control when called by subroutine 8.y, overlay 1.

Subroutine 5.x - Switches data in registers 0, 2, and 4 with registers 1, 3, and 5 respectively to differentiate firer data from target data.

Subroutine 6.x - Determines next weapon code and next overlay when called.

Subroutine 7.x - Calculates number of firers and targets acquired and returns to calling subroutine (0 or 1.x).

Subroutine 8.y - Calculates SSHP and engagement rate and returns to calling subroutine (0 or 1.x). Calls subroutine 3.y for execution control.

Figure 3-2. Overlay 1

Overlay 2

Submodel A - Calls subroutine 2.x and calculates targets hit by Blue tanks on return.

Submodel B - Allocates fires and calculates attrition. Calls subroutines 5.x and 6.x for execution control.

Subroutine 2.x - Calculates engagement rate for Blue tanks. Calls subroutine 5.x for execution control. Calls subroutine 7.x to calculate number of firers and targets acquired. Calls subroutine 8.x to calculate the SSHP for Blue tanks. Calls subroutine 9 to limit engagement rate.

Subroutine 5.x - Switches data in registers 0, 2, and 4 with registers 1, 3, and 5 respectively to differentiate firer data from target data.

Subroutine 6.x - Determines next weapon code and next overlay when called.

Subroutine 7.x - Calculates number of firers and targets acquired and returns to subroutine 2.x.

Subroutine 8.x - Calculates SSHP and returns to subroutine 2.x.

Subroutine 9 - Limits Blue tank engagement rate when called.

Figure 3-3. Overlay 2

Overlay 3

Submodel A - Calls subroutine 3.x and calculates targets hit by Red tanks on return.

Submodel B - Allocates fires and calculates attrition. Calls subroutines 5.x and 6.x for execution control.

Subroutine 3.x - Calculates engagement rate for Red tanks. Calls subroutine 5.x for execution control. Calls subroutine 7.x to calculate number of firers and targets acquired. Calls subroutine 8.x to calculate the SSHP for Red tanks. Calls subroutine 4.x to limit engagement rate.

Subroutine 4.x - Limits Red tank engagement rate when called.

Subroutine 5.x - Switches data in registers 0, 2, and 4 with registers 1, 3, and 5 respectively to differentiate firer data from target data.

Subroutine 6.x - Determines next weapon code and next overlay when called.

Subroutine 7.x - Calculates number of firers and targets acquired and returns to subroutine 3.x.

Subroutine 8.x - Calculates SSHP and returns to subroutine 3.x.

Figure 3-4. Overlay 3

Overlay 4

Submodel A and Subroutine 4.y - Calculates the probability of line of sight to targets in the coming time step.

Submodel B - Following each time step calculates the suppression of Red and Blue ATGMs in the succeeding time step. Calls subroutines 2.z and 3.z.

Submodel C - Following each time step calculates $B(t)/B_0$, $R(t)/R_0$, SFRD, LER, FER, current Battle Range and current Battle Time and calls subroutine 4.y to update PLOS.

Submodel D - On call returns all acquisition and movement rates for confirmation or update.

Submodel E - On call commits Blue or Red reinforcement and updates appropriate registers. Calls subroutine 1.y for execution control.

Subroutine 1.y - Switches data in registers 0, 2, and 4 with registers 1, 3, and 5 respectively for execution control.

Subroutine 2.z - Determines the suppression of Red ATGMs in the succeeding time step and returns. Calls subroutine 5.y to calculate the suppression level.

Subroutine 3.z - Determines the suppression of Blue ATGMs in the succeeding time step and returns. Calls subroutine 5.y to calculate the suppression level.

Subroutine 4.y - See Submodel A, overlay 4.

Subroutine 5.y - On call calculates the level of suppression for Red and Blue ATGMs and returns to subroutines 2.z and 3.z respectively.

Subroutines 6.y and 7.y - Provide execution control for submodel suppression.

Figure 3-5. Overlay 4

CHAPTER 4
USER INSTRUCTIONS

4-1. USER INSTRUCTIONS. User instructions for the four overlays of this HP-67 Combat Model are presented in the following pages.

User Instructions

Missile Hits and Kills - Overlay 1

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Read recorded data card			0.00
2	Read overlay 4 (first time step only)			
3	Compute probability of LOS for first range step		A	PLOS
4	Read overlay 1			
5	Compute Blue missile hits (Red targets)		A	Tgts Hit
6	Compute Blue missile hits on Red tanks		B	Tanks Hit
7	Compute Blue missile kills of Red tanks	P K/H	R/S	Tks Killed
	Compute Blue missile hits on Red ATGM		None	ATGM Hit
8	Compute Blue missile kills of Red ATGM	P K/H	R/S	ATGM Kills
	Program calls Red missile routine			1.00
	(same overlay)			1.00
9	Compute Red missile hits (Blue targets)		A	Tgts Hit
10	Compute Red missile hits on Blue tanks		B	Tanks Hit
11	Compute Red missile kills of Blue tanks	P K/H	R/S	Tks Killed
	Compute Red missile hits on Blue ATGM		None	ATGMs Hit
12	Compute Red missile kills of Blue ATGM	P K/H	R/S	ATGMs Killed
	Program calls next overlay for execution:			
	Will call overlay 4 if neither Blue nor Red			
	tank opens fire			
	Range has been reached			
	See overlay 4 instructions			
	Will call overlay 2 when Blue tank open fire			
	range is reached			
	See overlay 2 instructions			

User Instructions

Blue Tank - Overlay 2

Red Tank - Overlay 3

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1	Read overlay 2			
2	Targets hit by Blue tank		A	Tgts Hit
3	Tanks hit		B	Tank Hit
4	Tanks killed	P K/H	R/S	Tanks Killed
	ATGMs hit			ATGMs Hit
5		P K/H	R/S	ATGMs Killed
	Calls overlay 3			
	Do step 2 as in overlay 1		A	Tgts Hit
	If Red tank open fire range has not been reached 0 hits are flashed and overlay 4 is called.			
	If Red tank can fire hits are shown - do step 3, 4, 5 and program calls overlay 4.			

User Instructions

PLOS, Output, Update,
and Reserves - Overlay 4

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
0	Compute initial PLOS for time step 1		A	PLOS
	(unnecessary for remainder of battle)			
1	ATGM suppression		B	% Red ATGM
				not supp
				% Blue
				ATGM not
				supp
3	Battle outcomes and status update		C	% Blue Surv
	(calls overlay 1 on completion)			% Red Surv
				SFRD
				LER
				FER
				Battle Range
				Battle Time
4	(Optional) Recalls all acquisition:		D	Blue ATGM
	movement rates for update or confirmation			Acq Rate
			R/S	Red ATGM
				Acq Rate
			R/S	Blue Tank
				Acq Rate
			R/S	Red Tank
				Acq Rate
			R/S	Move Rate
				kph
			R/S	Range Step
				Next
				Overlay
5	(Optional) Reserve Commitment - Blue	0	E	$\alpha_2 R(t)$
	Reserve Commitment - Red	1	E	$\alpha_1 B(t)$
6	Input total reserve		R/S	$\Delta B(t)$
7	Input reserve ATGMs			$\Delta R(t)$
	Get new α_1 or α_2			α_1 or α_2
				Next
				Overlay

CHAPTER 5
PROGRAM LISTINGS

5-1. GENERAL. This chapter presents the program listings for this HP-67 Combat Model.

5-2. SUBMODEL ENGAGEMENT (Label A, Overlays 1, 2, and 3). Calls subroutines 0, 1.x, 2.x, and 3.x to calculate SSHP, number of firers, engagement rate, and targets acquired for each weapon code. Calculates targets hit on return.

```
LBL A
GSB(i)
1
RCL B
RCL A
÷
-
RCL C
RCL D
X
Y^X
1
-
CHS
RCL A
X
STO A
RTN (Targets Hit)
```

5-3. SUBMODEL LINE OF SIGHT (Labels A and 4, Overlay 4). Calculates the probability of LOS to targets in the coming time step.

```
LBL A
LBL 4
1.68
RCL 6
X
r̄ (1.2)
÷
CHS
e^X
LST X
CHS
a (.57)
+
X
STO 9 IPLOS)
RTN
```

5-4. SUBMODEL FIRE ALLOCATION AND ATTRITION (Label B, Overlays 1, 2, and 3). Calculates fire allocation and computes targets killed by type for each weapon code. Calls subroutines 2.x (overlay 1) and 6.x. Contains subroutine 5.x which controls execution along with subroutines 2.x and 6.x.

LBL B
RCL 1
RCL 2
-
RCL 5
X
STO 5
RCL 6
3
y^x
a_1 (.01)
X
CHS
RCL 6
+
X
a_2 (.05)
X
-
RCL 6
a_3 (.04)
X
-
1
+
RCL A
X

R/S (Tanks Hit)
Input P(K/H)
X
STO + 2
-X- (Tanks Killed)
LST X
+
RCL A
-
CHS
R/S (ATGMs Hit)
Input P(K/H)
X
STO + 2
-X- (ATGMs Killed)
RCL 5
-
RCL 2
RCL 1
-
+
STO 5
LBL 5
RCL 0
RCL 1
STO 0

R_v
STO 1
RCL 2
RCL 3
STO 2
R_v
STO 3
RCL 4
RCL 5
STO 4
R_v
STO 5
RCL I
3
x = y
GTO 6
$x \lessgtr y$
1
+
2 } Overlay 1 Only
x = y }
GTO 2 }
LST X }
STO I
PAUSE (Weapon Code)
RTN (Next Overlay)

5-5. SUBMODEL SUPPRESSION (Label B, Overlay 4).

```
LBL B
0
RCL 8
x > y
GTO 2
1
STO 8
PAUSE
SF 2
GTO 3
LBL 2
RCL 8
RCL 5
RCL 1
RCL 4
-
X
÷
LBL 5
↑
↑
1
-
2
y^x
x ≶ y
÷
.04
X
CHS
p (1.0)
÷
e^x
10
X
5
-
e^x
↑
↑
1
+
÷
1
-
CHS
F?2
GTO 6
STO 8
PAUSE
SF 2
LBL 3
p ≶ s
RCL 0
p ≶ s
RCL 6
x > y
GTO 7
RCL C
RCL 4
RCL 0
RCL 3
-
X
÷
GTO 5
LBL 6
STO C
RTN
LBL 7
1
STO C
CF 2
RTN
```

5-6. SUBMODEL OUTPUT (Label C, Overlay 4). Calculates output data each time step and updates battle range and battle time. Calls subroutine 4.y to update LOS probability.

```
LBL C
1
RCL 3
RCL 0
÷
-
-X- (B(t)/B0)
1
RCL 2
RCL 1
÷
-
-X- (R(t)/R0)
-
-X- (SFRD)
RCL 2
RCL 3
÷
-X- (LER)
RCL 1
RCL 0
÷
÷
-X- (FER)
RCL 6
RCL E
-
STO 6
-X- (Battle Range)
GSB 4
RCL 7
1
+
STO 7
-X- (Battle Time)
1
RTN (Next Overlay)
```

5-7. SUBMODEL UPDATE (Label D, Overlay 4). On call updates acquisition and movement rates and calculates the range step and displays current data. When program execution stops either accept the current value and key run or key in updated input and key run.

LBL D

p $\lessgtr$ s

RCL 6

R/S (Blue ATGM Acquisition Rate)

STO 6

RCL 7

R/S (Red ATGM Acquisition Rate)

STO 7

RCL 8

R/S (Blue Tank Acquisition Rate)

STO 8

RCL 9

R/S (Red Tank Acquisition Rate)

STO 9

p $\lessgtr$ s

RCL E

60

X

R/S (Movement Rate in kph)

60

÷

STO E

PAUSE (Range Step in km)

1

RTN (Next Overlay)

5-8. SUBMODEL RESERVE COMMITMENT (Label E, Overlay 4). On call commits either the Red or Blue reserve and updates appropriate register. Contains subroutine 1.y for execution control.

Enter 0 = Blue or 1 = Red from key board.

LBL E
STO B
1
x = y
GSB 1
RCL 4
RCL 0
RCL 3
-
X
STO A
R/S (α_1 B(t) or α_2 R(t))
Input Total Reserve
STO + 0
R/S (Δ B(t) or Δ R(t))
Input Reserve ATGMs
$x \lessgtr y$
÷

LST X
X
RCL A
+
RCL 0
RCL 3
-
÷
STO 4
-X- (updated α_1 or α_2)
RCL B
1
x = y
GSB 1
RTN (Next Overlay)
LBL 1
RCL 0

RCL 1
STO 0
R_v
STO 1
RCL 2
RCL 3
STO 2
R_v
STO 3
RCL 4
RCL 5
STO 4
R_v
STO 5
1
RTN (Next Overlay)

5-9. SUBROUTINES.

a. <u>0 - Blue Missile, Overlay 1</u>. Calls subroutine 8.y to calculate SSHP and engagement rate. Calls subroutine 7 to calculate targets acquired and number of firers.

```
LBL 0
p ≶ s
RCL 6
GSB 7
RCL C
X
STO C
GSB 8
RTN
```

b. <u>1.x - Red Missile, Overlay 1</u>. Calls subroutine 8.y to calculate SSHP and engagement rate. Calls subroutine 7 to calculate targets acquired and number of firers.

```
LBL 1
p ≶ s
RCL 7
GSB 7
RCL 8
X
STO C
GSB 8
RTN
```

c. <u>1.y - Data switch for execution control</u>. See LBL 1, Submodel E, Overlay 4.

d. <u>2.x - Blue Tank, Overlay 2</u>. Calculates engagement rate for Blue tanks. Calls subroutine 7 to calculate targets acquired and number of firers and calls subroutine 8.x to calculate SSHP. Contains subroutine 9 to set maximum engagement rate. Calls subroutine 5 for execution control.

```
LBL 2
p ≶ s
RCL 1
STO D
p ≶ s
RCL 6
x > y
GTO 5
p ≶ s
RCL 8
GSB 7
p ≶ s
RCL 5
STO B
RCL 4
GSB 8
RCL D
Depth of Defense
 in km
-
RCL 6
x ≤ y
GTO 9
RCL D
RCL 6
RCL E
-
-
Maximum Engagement
 rate in rounds/min
X
Depth of Defense
 in km
+
STO D
RTN
LBL 9
Maximum Engagement
 rate in rounds/min
STO D
RTN
```

e. 2.y - Execution Control, Overlay 1 Only.

```
LBL 2
p ≶ s
RCL 1
p ≶ s
RCL 6
x > y
GTO 6
2
STO I
PAUSE (Weapon Code)
RTN (Next Overlay)
```

f. 2.z - Red ATGM Suppression. See Label 2, Submodel Suppression = Label B of Overlay 4.

g. 3.x - Red Tank, Overlay 3. Calculates engagement rate for Red tanks. Calls subroutine 7 to calculate targets acquired and number of firers and calls subroutine 8.x to calculate SSHP. Contains subroutine 4 to set maximum engagement rate. Calls subroutine 5 for execution control.

```
LBL 3
p ≶ s
RCL 0
STO D
p ≶ s
RCL 6
x > y
GTO 5
p ≶ s
RCL 9
GSB 7
p ≶ s
RCL 3
STO B
RCL 2
GSB 8
RCL D
Depth of Attack
 Formation in km
-
RCL 6
x ≤ y
GTO 4
Maximum Engagement
 Rate in rounds/min
1
+
ln X
Depth of Attack
 Formation in km
÷
RCL 6
RCL E
-
RCL D
-
X
CHS
e^x
1
-
STO D
RTN
LBL 4
Maximum Engagement
 Rate in rounds/min
STO D
RTN
```

h. 3.z - Blue ATGM Suppression. See Label 3, Submodel Suppression, Label B of Overlay 4.

i. 3.y - Missile Engagement Control. See Label 3, Subroutine 8.y.

j. 4.x - Red Tank Engagement Control. See Label 4, Subroutine 3.

k. 4.y - Subroutine Line of Sight. See Submodel A, Overlay 4.

l. 5.x - Data Switch for Execution Control. See Label 5, Submodel Fire Allocation and Attrition.

m. 6.x - Execution Control, Overlays 1,2, and 3.

```
LBL 6                        4
0                            RTN (Next Overlay)
STO I                          (Overlay 1 and 2)
PAUSE (Weapon Code)          R/S (Next Overlay)
                               (Overlay 3)
```

n. 5.y - Red and Blue ATGM Suppression. See Label 5, Submodel Suppression, Label B of Overlay 4.

o. 6.y and 7.y - Execution Control for Submodel Suppression. See Labels 6 and 7 of Submodel B, Overlay 4.

p. 7.x - Target Acquisition, Overlays 1, 2, and 3.

```
LBL 7                        1 } Overlay 1 Only
CHS                          + }
e^x                          STO A (Targets Acquired)
1
-                            Overlay 1:        Overlays 2 and 3:
CHS                            RCL 4             1
p ≶ s                          RCL 0             RCL 4
RCL 9                          RCL 3             -
X                                                RCL 0
RCL 1                                            RCL 3
RCL 2                        -
-                            X
X                            {STO C} (Number of Firers)
                                     (Overlays 2 and 3 only)
                             RTN
```

q. <u>8.x - Tank SSHP Computation, Overlays 2 and 3.</u>

```
LBL 8        RCL B
p ≶ s        X
RCL 6        +
↑            1
X            +
X            STO B
RCL 6        RTN
```

r. <u>8.y - AT Missile SSHP and Engagement Rate Computation, Overlay 1.</u>

```
LBL 8        STO D
RCL I        RTN
1            LBL 3
x = y        RCL 6
GTO 3        ln X
RCL 6        .2
ln X         X
.09          .7
X            +
.84          STO B
+            RCL 6
STO B        .07
RCL 6        X
.2           CHS
X            .74
CHS          +
1.4          STO D
+            RTN
```

s. <u>9 - Blue Tank Engagement Control</u>. See Label 9, Subroutine 2.

5-10. HP-67 COMBAT MODEL STORAGE USE.

a. Primary Registers.

[1]A	Targets Acquired	2	Σ r(t)
[2]B	SSHP	3	Σ b(t)
[3]C	Number of Firers	4	α_1
D	Engagement Rate	[4]5	α_2
E	Range Step	6	Battle Range
I	Current Weapon Code	7	Battle Time
0	B_0	8	Red ATGMs Killed Last Δt
1	R_0	9	PLOS

b. Protected Secondary Registers.

0 Red Tank Open Fire Range
1 Blue Tank Open Fire Range
2 Red Tank SSHP Parameter b
3 Red Tank SSHP Parameter a
4 Blue Tank SSHP Parameter b
5 Blue Tank SSHP Parameter a
6 Blue ATGM Acquisition Rate
7 Red ATGM Acquisition Rate
8 Blue Tank Acquisition Rate
9 Red Tank Acquisition Rate

---

[1] Also used for targets hit in Submodel Engagement and α_1 B(t) or α_2 B(t) in Submodel Reserve.

[2] Also used to index Red or Blue reinforcement in Submodel Reserve.

[3] Also used for Blue ATGMs killed last Δt.

[4] Also used for opposing ATGMs last Δt in Submodel Allocation and Attrition - α_1 recomputed and restored.

DISTRIBUTION

HQDA (SAUS-OR - Mr Hardison)
Rm 2E 621
Pentagon
Washington, DC 20310

Director
US Army TRADOC Systems Analysis Activity
ATTN: ATAA-D (Dr Payne)
White Sands Missile Range, NM 88002

Dr Seth Bonder
Vector Research Inc
PO Box 1506
Ann Arbor, MI 48106

Commander
US Army Training and Doctrine Command
ATTN: ATCD-AO (LTC(P) Pokorny)
Fort Monroe, VA 23651

DCDR, CACDA
ATTN: ATCA-DC
Fort Leavenworth, Kansas 66027

Commander
US Army Combat Developments Experimentation Command
ATTN: Dr Marion Bryson, SA
Ft Ord, CA 93941

Hewlett Packard Program Library
19310 Pruneridge Avenue
Cupertino, California 95014